EXAMPLE 9b (part I): intermediate demand on own products

	Production Sectors				Consumers
Markets	X	Y	W	own	CONS
PX	100+20		-100	-20	
PY		100+20	-100	-20	
PW			200		-200
PL	-40	-60			100
PK	-60	-40			100
own demand	-20	-20		40	

The column "own" is just needed to balance the matrix, but there is no need to distinguish it in the model.

The production function is represented by nested CES-Cobb-Douglass function:

$X = A * [\delta X^{(\sigma-1)/\sigma} + (1-\delta) f(K,L)^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

$Y = A * [\delta Y^{(\sigma-1)/\sigma} + (1-\delta) f(K,L)^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

where $f(K,L) = A * K^{\alpha} * L^{1-\alpha}$

```
$ONTEXT
$MODEL:M1_2I

$SECTORS:
        X       ! Activity level for sector X
        Y       ! Activity level for sector Y
        W       ! Activity level for sector W (Hicksian welfare index)

$COMMODITIES:
        PX      ! Price index for commodity X
        PY      ! Price index for commodity Y
        PL      ! Price index for primary factor L
        PK      ! Price index for primary factor K
        PW      ! Price index for welfare (expenditure function)

$CONSUMERS:
        CONS    ! Income level for consumer CONS

$PROD:X s:0.5   va:1
        O:PX    Q:120
        I:PX    Q:20
        I:PL    Q:40    va:
        I:PK    Q:60    va:

$PROD:Y s:0.75  va:1
        O:PY    Q:120
        I:PY    Q:20
        I:PL    Q:60    va:
        I:PK    Q:40    va:

$PROD:W s:1
        O:PW    Q:200
        I:PX    Q:100
        I:PY    Q:100

$DEMAND:CONS
        D:PW    Q:200
        E:PL    Q:100
        E:PK    Q:100

$OFFTEXT
$SYSINCLUDE mpsgeset M1_2I

        M1_2I.ITERLIM = 0;
$INCLUDE M1_2I.GEN
        SOLVE M1_2I USING MCP;
```





EXAMPLE 9b (part II): non-symmetric intermediate demand

	Production Sectors				Consumers
Markets	X	Y	W	own	CONS
PX	100+20+5		-100-5	-20	
PY	-5	100+20	-100+5	-20	
PW			200		-200
PL	-40	-60			100
PK	-60	-40			100
own demand	-20	-20		40	

Modifications in X production structure requires to modify W structure.

The production function is represented by nested CES-Cobb-Douglass function:

$X = A * [\delta X^{(\sigma-1)/\sigma} + \beta Y^{(\sigma-1)/\sigma} + (1-\delta-\beta) f(K,L)^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

$Y = A * [\delta Y^{(\sigma-1)/\sigma} + (1-\delta) f(K,L)^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

where $f(K,L) = A * K^{\alpha} * L^{1-\alpha}$

```
$ONTEXT
$MODEL:M1_2II

...

$PROD:X
s:0.5   va:1
        O:PX    Q:125
        I:PX    Q:20
        I:PY    Q:5
        I:PL    Q:40    va:
        I:PK    Q:60    va:

$PROD:Y s:0.75  va:1
        O:PY    Q:120
        I:PY    Q:20
        I:PL    Q:60    va:
        I:PK    Q:40    va:

$PROD:W s:1
        O:PW    Q:200
        I:PX    Q:105
        I:PY    Q:95

$DEMAND:CONS
        D:PW    Q:200
        E:PL    Q:100
        E:PK    Q:100

$OFFTEXT
$SYSINCLUDE mpsgeset M1_2II

        M1_2II.ITERLIM = 0;
$INCLUDE M1_2II.GEN
        SOLVE M1_2II USING MCP;
```



Y remains as before

EXAMPLE 9b (part III): intermediate demand

	Production Sectors				Consumers
Markets	X	Y	W	own	CONS
PX	100+20+5	-7	-100-5+7	-20	
PY	-5	100+20+7	-100+5-7	-20	
PW			200		-200
PL	-40	-60			100
PK	-60	-40			100
own demand	-20	-20		40	

The production function is represented by nested CES-Cobb-Douglass function:

$X = A * [\delta X^{(\sigma-1)/\sigma} + \beta Y^{(\sigma-1)/\sigma} + (1-\delta-\beta) f(K,L)^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

$Y = A * [\delta X^{(\sigma-1)/\sigma} + \beta Y^{(\sigma-1)/\sigma} + (1-\delta) f(K,L)^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

where $f(K,L) = A * K^{\alpha} * L^{1-\alpha}$

```
$ONTEXT
$MODEL:M1_2III

...

$PROD:X s:0.5   va:1
        O:PX    Q:125
        I:PX    Q:20
        I:PY    Q:5
        I:PL    Q:40    va:
        I:PK    Q:60    va:

$PROD:Y s:0.75  va:1
        O:PY    Q:127
        I:PX    Q:7
        I:PY    Q:20
        I:PL    Q:60    va:
        I:PK    Q:40    va:

$PROD:W s:1
        O:PW    Q:200
        I:PX    Q:98
        I:PY    Q:102

$DEMAND:CONS
        D:PW    Q:200
        E:PL    Q:100
        E:PK    Q:100

$OFFTEXT
$SYSINCLUDE mpsgeset M1_2III

        M1_2III.ITERLIM = 0;
$INCLUDE M1_2III.GEN
        SOLVE M1_2III USING MCP;
```



X remains as before

Conclusion: (i) MPSGE allows for any input-output structure. **(ii)** The benchmark equilibrium always require to design a model in such way that initial data will be replicated. MPSGE by default set all variables (quantity and prices, but not values) equal to 1 (that can be interpreted as a multipliers). This means that there is no sense to compare results between different benchmark equilibriums.

Exercise 9b_A:

Implement a separate nest between inputs X and Y in the same way as between K and L.

The production function is represented by nested CES-Cobb-Douglass function:

$X = A * [\delta f(X,Y)^{(\sigma-1)/\sigma} + (1-\delta) f(K,L)^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

$Y = A * [\delta f(X,Y)^{(\sigma-1)/\sigma} + (1-\delta) f(K,L)^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

where $f(K,L) = A * K^{\alpha} * L^{1-\alpha}$

$f(X,Y) = A * [\delta X^{(\sigma-1)/\sigma} + (1-\delta) Y^{(\sigma-1)/\sigma}]^{\sigma/(\sigma-1)}$

```
$PROD:X s:0.5    va:1   vb:0.9
        O:PX     Q:125
        I:PX     Q:20     vb:
        I:PY     Q:5      vb:
        I:PL     Q:40     va:
        I:PK     Q:60     va:
```




```
$PROD:Y s:0.75   va:1    vc:1.2
        O:PY     Q:127
        I:PX     Q:7      vc:
        I:PY     Q:20     vc:
        I:PL     Q:60     va:
        I:PK     Q:40     va:
```



Conclusion: MPSGE allows for any combinations of nested functions

Exercise 9b_B:

Implement the same tax as in EXAMPLE 9a into the original model M1_2III and into its modified version from Exercise 9b_A.

Original model

```
SCALAR TX  /0/;
$ONTEXT
$MODEL:M1_2III

...
$PROD:X s:0.5   va:1
        O:PX    Q:125
        I:PX    Q:20
        I:PY    Q:5
        I:PL    Q:40    va:      A:CONS T:TX
        I:PK    Q:60    va:      A:CONS T:TX

PROD:Y s:0.75  va:1
        O:PY    Q:127
        I:PX    Q:7
        I:PY    Q:20
        I:PL    Q:60    va:
        I:PK    Q:40    va:
...
$REPORT:
       V:YY  O:PY   PROD:Y
       V:XX  O:PX   PROD:X
       V:XC  I:PX   PROD:W
       V:YC  I:PY   PROD:W
$OFFTEXT
$SYSINCLUDE mpsgeset M1_2III
M1_2III.ITERLIM = 0;
$INCLUDE M1_2III.GEN
        SOLVE M1_2III USING MCP;
TX = 1.0;
M1_2III.ITERLIM = 1000;
$INCLUDE M1_2III.GEN
        SOLVE M1_2III USING MCP;
```

```
            LOWER      LEVEL      UPPER

---- VAR X        .        0.706     +INF
---- VAR Y        .        1.256     +INF
---- VAR W        .        0.948     +INF
---- VAR PX       .        2.103     +INF
---- VAR PY       .        1.158     +INF
---- VAR PL       .        1.163     +INF
---- VAR PK       .        1.026     +INF
---- VAR PW       .        1.551     +INF
---- VAR CONS     .      294.142     +INF
---- VAR YY       .      159.465     +INF
---- VAR XX       .       88.271     +INF
---- VAR XC       .       68.531     +INF
---- VAR YC       .      129.593     +INF
```

Modified version

```
SCALAR TX  /0/;
$ONTEXT
$MODEL:M1_2III

...
$PROD:X s:0.5   va:1      vb:0.9
        O:PX    Q:125
        I:PX    Q:20      vb:
        I:PY    Q:5       vb:
        I:PL    Q:40      va:      A:CONS T:TX
        I:PK    Q:60      va:      A:CONS T:TX

PROD:Y s:0.75  va:1
        O:PY    Q:127
        I:PX    Q:7
        I:PY    Q:20
        I:PL    Q:60    va:
        I:PK    Q:40    va:
...
$REPORT:
       V:YY  O:PY   PROD:Y
       V:XX  O:PX   PROD:X
       V:XC  I:PX   PROD:W
       V:YC  I:PY   PROD:W
$OFFTEXT
$SYSINCLUDE mpsgeset M1_2III
M1_2III.ITERLIM = 0;
$INCLUDE M1_2III.GEN
        SOLVE M1_2III USING MCP;
TX = 1.0;
M1_2III.ITERLIM = 1000;
$INCLUDE M1_2III.GEN
        SOLVE M1_2III USING MCP;
```

```
                LOWER      LEVEL      UPPER

---- VAR X          .        0.701     +INF
---- VAR Y          .        1.262     +INF
---- VAR W          .        0.947     +INF
---- VAR PX         .        2.102     +INF
---- VAR PY         .        1.160     +INF
---- VAR PL         .        1.167     +INF
---- VAR PK         .        1.027     +INF
---- VAR PW         .        1.553     +INF
---- VAR CONS       .      294.142     +INF
---- VAR YY         .      160.236     +INF
---- VAR XX         .       87.586     +INF
---- VAR XC         .       68.553     +INF
---- VAR YC         .      129.292     +INF
```

The tax on inputs for X implies that production of X becomes more expensive:

consumer price $\Delta Pd = PX_2 - PX_1 = 2.103 - 1 = 1.103$

producer price $\Delta Ps = PX_2/(1+TX) - PX_1 = 2.103/2 - 1 = 0.05$ (this means that producer shift the whole tax burden on consumers)



Y becomes more attractive for consumers due to more expensive X => X and Y are substitute goods
The nesting structure in original model can be represented in two ways:

Original model		Modified model
X=125 s=0.5 KL X=20 Y=5 va=1 L =40 K=60	X=125 s=0.5 KL XY va=1 s=0.5 L=40 K=60 X=20 Y=5	X=125 s=0.5 KL XY va=1 vb=0.9 L=40 K=60 X=20 Y=5

The only difference between both models is the elasticity of substitution between X and Y for production process of X. Since the elasticity ↑ ⇒ producer X can easily substitute X and Y than before ⇒ better technology ⇒ PX↓ ⇒ production of Y and X becomes cheaper since both sectors use X as an input ⇒ Y↑ ⇒ demand on K↑ and L↑ ⇒ PK↑and PL↑, and PK↑< PL↑ since sector Y is more labour intensive, i.e. this sector increase more demand on L than on K ⇒ PY↑ ⇒ X↓

Households welfare decreases from U_1 to U_2 due to PX/PY↓ since PX↓ and PY↑



Conclusion: (i) Changes in nested structure of production function may affect welfare **(ii)** Innovations, like technology improvement, should theoretically improve households welfare. The above example shows that this is not always the case.

Exercise 9b_C:

Revise the model M1_2III by using another calibration point (see EXAMPLE 1 on Lecture 2b)

The demand function in original model can be represented in two ways:

ORIGINAL		MODIFIED	
A	A	C	C
$ONTEXT $MODEL:M1_2III ... $PROD:W s:1 O:PW Q:200 I:PX Q:98 I:PY Q:102 $DEMAND:CONS D:PW Q:200 E:PL Q:100 E:PK Q:100 $REPORT: V:XD I:PX PROD:W V:YD I:PY PROD:W $OFFTEXT $SYSINCLUDE ... PX.L=1; PY.L=1; ...	$ONTEXT $MODEL:M1_2III ... $DEMAND:CONS s:1 E:PL Q:100 E:PK Q:100 D:PX Q:98 D:PY Q:102 $REPORT: V:XD D:PX DEMAND:CONS V:YD D:PY DEMAND:CONS $OFFTEXT $SYSINCLUDE	$ONTEXT $MODEL:M1_2III ... $DEMAND:CONS s:1 E:PL Q:100 E:PK Q:100 D:PX Q:392 P:(1/4) D:PY Q:102 $REPORT: ... $OFFTEXT $SYSINCLUDE	ONTEXT $MODEL:M1_2III ... $DEMAND:CONS s:1 E:PL Q:100 E:PK Q:100 D:PX Q:98 P:(1/4) D:PY Q:25.5 $REPORT: ... $OFFTEXT $SYSINCLUDE ... ...

Let's assume MRS changes from 1 to ¼. How to modify the calibration point in order to obtain the same results as in the original version?

MRS(98,102)=1 - the original calibration point
MRS(?,102)=(98,?) =1/4- the modified calibration point

Since s:1, we have Cobb-Douglas utility function $U = aX^BY^C$
$MRS = \frac{bY}{cX} = 1 = \frac{b*102}{c*98}$ => b=98, c=102 => $\boldsymbol{U = X^{98}Y^{102}}$ or $\boldsymbol{U = X^{0.98}Y^{1.02}}$

If MRS(?,102)=1/4 => $MRS = \frac{0.98*102}{1.02*X} = \frac{1}{4}$ => X=392
If MRS(98,?) =1/4 => $MRS = \frac{0.98*Y}{1.02*98} = \frac{1}{4}$ => Y=25.5
If MRS(?,?) =1/4 => $MRS = \frac{0.98*Y}{1.02*X} = \frac{1}{4}$ => any solution that satisfy $\frac{Y}{X} = 0.26$ (e.g. X=100, Y=26)

All the above modifications will not affect the results unless we will not change price normalization from PX.L=1 to PX.L=1/4

MODIFIED

B

```
$ONTEXT
$MODEL:M1_2III
 ...

$DEMAND:CONS  s:1
E:PL     Q:100
E:PK     Q:100
D:PX     Q:392 P:(1/4)
D:PY     Q:102


$REPORT:
V:XD  D:PX  DEMAND:CONS
V:YD  D:PY  DEMAND:CONS

$OFFTEXT
$SYSINCLUDE ...

PX.L=1/4;
PY.L=1;
...
```

```
                LOWER   LEVEL      UPPER   MARGINAL

---- VAR X        .     1.000      +INF      74.550
---- VAR Y        .     1.000      +INF      -8.005
---- VAR PX       .     0.250      +INF    -322.655
---- VAR PY       .     1.000      +INF       1.353
---- VAR PL       .     1.000      +INF       6.059
---- VAR PK       .     1.000      +INF       6.706
---- VAR CONS     .   200.000       +INF        .
---- VAR XD       .   392.000       +INF        .
---- VAR YD       .   102.000       +INF        .
```

C

```
ONTEXT
$MODEL:M1_2III
 ...

$DEMAND:CONS  s:1
E:PL     Q:100
E:PK     Q:100
D:PX     Q:392 P:(1/4)
D:PY     Q:102


$REPORT:
...


$OFFTEXT
$SYSINCLUDE ...

PX.L=1;
PY.L=1;
...
```

```
                LOWER      LEVEL

---- VAR X        .         1.000
---- VAR Y        .         1.000
---- VAR PX       .         1.000
---- VAR PY       .         1.000
---- VAR PL       .         1.000
---- VAR PK       .         1.000
---- VAR CONS     .       200.000
---- VAR XD       .        98.000
---- VAR YD       .       102.000
```



Conclusion: (i) The change in calibration point supported by the appropriate change in (i) MRS + (ii) price normalization, will effect a change in equilibrium point. The change in calibration point supported by the appropriate change in MRS only, will keep the equilibrium point constant. **(ii)** If the price normalisation is differentiated (no matter of calibration point), the equilibrium level is biased (too big marginal value)